
NAG Library Routine Document

M01DEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

M01DEF ranks the rows of a matrix of real numbers in ascending or descending order.

2 Specification

SUBROUTINE M01DEF (RM, LDM, M1, M2, N1, N2, ORDER, IRANK, IFAIL)

INTEGER LDM, M1, M2, N1, N2, IRANK(M2), IFAIL
REAL (KIND=nag_wp) RM(LDM,N2)
CHARACTER(1) ORDER

3 Description

M01DEF ranks rows M1 to M2 of a matrix, using the data in columns N1 to N2 of those rows. The
ordering is determined by first ranking the data in column N1, then ranking any tied rows according to
the data in column N1þ 1, and so on up to column N2.

M01DEF uses a variant of list-merging, as described on pages 165–166 in Knuth (1973). The routine
takes advantage of natural ordering in the data, and uses a simple list insertion in a preparatory pass to
generate ordered lists of length at least 10. The ranking is stable: equal rows preserve their ordering in
the input data.

4 References

Knuth D E (1973) The Art of Computer Programming (Volume 3) (2nd Edition) Addison–Wesley

5 Arguments

1: RMðLDM;N2Þ – REAL (KIND=nag_wp) array Input

On entry: columns N1 to N2 of rows M1 to M2 of RM must contain real data to be ranked.

2: LDM – INTEGER Input

On entry: the first dimension of the array RM as declared in the (sub)program from which
M01DEF is called.

Constraint: LDM � M2.

3: M1 – INTEGER Input

On entry: the index of the first row of RM to be ranked.

Constraint: M1 > 0.

4: M2 – INTEGER Input

On entry: the index of the last row of RM to be ranked.

Constraint: M2 � M1.

5: N1 – INTEGER Input

On entry: the index of the first column of RM to be used.

Constraint: N1 > 0.

6: N2 – INTEGER Input

On entry: the index of the last column of RM to be used.

Constraint: N2 � N1.

7: ORDER – CHARACTER(1) Input

On entry: if ORDER ¼ A , the rows will be ranked in ascending (i.e., nondecreasing) order.

If ORDER ¼ D , into descending order.

Constraint: ORDER ¼ A or D .

8: IRANKðM2Þ – INTEGER array Output

On exit: elements M1 to M2 of IRANK contain the ranks of the corresponding rows of RM. Note
that the ranks are in the range M1 to M2: thus, if the ith row of RM is the first in the rank order,
IRANKðiÞ is set to M1.

9: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, M2 < 1,
or N2 < 1,
or M1 < 1,
or M1 > M2,
or N1 < 1,
or N1 > N2,
or LDM < M2.

IFAIL ¼ 2

On entry, ORDER is not `A' or `D'.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

M01DEF is not threaded in any implementation.

9 Further Comments

The average time taken by the routine is approximately proportional to n� log nð Þ, where
n ¼ M2�M1þ 1.

10 Example

This example reads a matrix of real numbers and ranks the rows in ascending order.

10.1 Program Text

Program m01defe

! M01DEF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: m01def, nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6

! .. Local Scalars ..
Integer :: i, ifail, j, ldm, m1, m2, n1, n2

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: rm(:,:)
Integer, Allocatable :: irank(:)

! .. Executable Statements ..
Write (nout,*) ’M01DEF Example Program Results’

! Skip heading in data file
Read (nin,*)

Read (nin,*) m2, n2
ldm = m2
Allocate (rm(ldm,n2),irank(m2))

m1 = 1
n1 = 1

Do i = m1, m2
Read (nin,*)(rm(i,j),j=n1,n2)

End Do

ifail = 0

Call m01def(rm,ldm,m1,m2,n1,n2,’Ascending’,irank,ifail)

Write (nout,*)
Write (nout,*) ’Data Ranks’
Write (nout,*)

Do i = m1, m2
Write (nout,99999)(rm(i,j),j=n1,n2), irank(i)

End Do

99999 Format (1X,3F7.1,I11)
End Program m01defe

10.2 Program Data

M01DEF Example Program Data
12 3
6.0 5.0 4.0
5.0 2.0 1.0
2.0 4.0 9.0
4.0 9.0 6.0
4.0 9.0 5.0
4.0 1.0 2.0
3.0 4.0 1.0
2.0 4.0 6.0
1.0 6.0 4.0
9.0 3.0 2.0
6.0 2.0 5.0
4.0 9.0 6.0

10.3 Program Results

M01DEF Example Program Results

Data Ranks

6.0 5.0 4.0 11
5.0 2.0 1.0 9
2.0 4.0 9.0 3
4.0 9.0 6.0 7
4.0 9.0 5.0 6
4.0 1.0 2.0 5
3.0 4.0 1.0 4
2.0 4.0 6.0 2
1.0 6.0 4.0 1
9.0 3.0 2.0 12
6.0 2.0 5.0 10
4.0 9.0 6.0 8

